
Overview: REG Procedure
The REG procedure is one of many regression procedures in the SAS System. It is a general-purpose
procedure for regression, while other SAS regression procedures provide more specialized applications.

Other SAS/STAT procedures that perform at least one type of regression analysis are the CATMOD, GEN-
MOD, GLM, LOGISTIC, MIXED, NLIN, ORTHOREG, PROBIT, RSREG, and TRANSREG procedures.
SAS/ETS procedures are specialized for applications in time series or simultaneous systems. These other
SAS/STAT regression procedures are summarized in Chapter 4, “Introduction to Regression Procedures,”
which also contains an overview of regression techniques and defines many of the statistics computed by
PROC REG and other regression procedures.

PROC REG provides the following capabilities:

� multiple MODEL statements

� nine model-selection methods

� interactive changes both in the model and the data used to �t the model

� linear equality restrictions on parameters

� tests of linear hypotheses and multivariate hypotheses

� collinearity diagnostics

� predicted values, residuals, studentized residuals, con�dence limits, and in�uence statistics

� correlation or crossproduct input

� requested statistics available for output through output data sets

� ODS Graphics. For more information, see the section “ODS Graphics” on page 8089.

Nine model-selection methods are available in PROC REG. In the simplest method, PROC REG �ts the
complete model that you specify. The other eight methods involve various ways of including or excluding
variables from the model. You specify these methods with the SELECTION= option in the MODEL statement.

The methods are identi�ed in the following list and are explained in detail in the section “Model-Selection
Methods” on page 8033.

NONE no model selection. This is the default. The complete model speci�ed in the MODEL
statement is �t to the data.

FORWARD forward selection. This method starts with no variables in the model and adds variables.

BACKWARD backward elimination. This method starts with all variables in the model and deletes
variables.

STEPWISE stepwise regression. This is similar to the FORWARD method except that variables
already in the model do not necessarily stay there.

MAXR forward selection to �t the best one-variable model, the best two-variable model, and so
on. Variables are switched so that R square is maximized.

MINR similar to the MAXR method, except that variables are switched so that the increase in R
square from adding a variable to the model is minimized.

RSQUARE �nds a speci�ed number of models with the highest R square in a range of model sizes.

ADJRSQ �nds a speci�ed number of models with the highest adjusted R square in a range of model
sizes.

CP �nds a speci�ed number of models with the lowestCp in a range of model sizes.

Getting Started: REG Procedure

Simple Linear Regression

Suppose that a response variableY can be predicted by a linear function of a regressor variableX. You can
estimate� 0, the intercept, and� 1, the slope, in

Yi D � 0 C � 1X i C � i

for the observationsi D 1; 2; : : : ; n. Fitting this model with the REG procedure requires only the following
MODEL statement, wherey is the outcome variable andx is the regressor variable.

proc reg;
model y=x;

run;

For example, you might use regression analysis to �nd out how well you can predict a child's weight if you
know that child's height. TheClass data set used in this example is available in theSashelp library.

The equation of interest is

Weight D � 0 C � 1Height C �

The variableWeight is the response or dependent variable in this equation, and� 0 and� 1 are the unknown
parameters to be estimated. The variableHeight is the regressor or independent variable, and� is the unknown
error. The following commands invoke the REG procedure and �t this model to the data.

ods graphics on;

proc reg data=sashelp.class;
model Weight = Height;

run;

Figure 99.1 includes some information concerning model �t.

TheF statistic for the overall model is highly signi�cant (F = 57.076,p < 0.0001), indicating that the model
explains a signi�cant portion of the variation in the data.

The degrees of freedom can be used in checking accuracy of the data and model. The model degrees of
freedom are one less than the number of parameters to be estimated. This model estimates two parameters,
� 0 and� 1; thus, the degrees of freedom should be 2 – 1 = 1. The corrected total degrees of freedom are
always one less than the total number of observations in the data set, in this case 19 – 1 = 18.

Several simple statistics follow the ANOVA table. The Root MSE is an estimate of the standard deviation of
the error term. The coef�cient of variation, or Coeff Var, is a unitless expression of the variation in the data.
The R-square and Adj R-square are two statistics used in assessing the �t of the model; values close to 1
indicate a better �t. The R-square of 0.77 indicates thatHeight accounts for 77% of the variation inWeight.

Figure 99.1 ANOVA Table

The “Parameter Estimates” table in Figure 99.2 contains the estimates of� 0 and� 1. The table also contains
thet statistics and the correspondingp-values for testing whether each parameter is signi�cantly different
from zero. Thep-values (t = –4.43,p = 0.0004 andt = 7.55,p < 0.0001) indicate that the intercept andHeight
parameter estimates, respectively, are highly signi�cant.

From the parameter estimates, the �tted model is

Weight D � 143:0C 3:9� Height

Figure 99.2 Parameter Estimates

If ODS Graphics is enabled, then PROC REG produces a variety of plots. Figure 99.3 shows a plot of the
residuals versus the regressor and Figure 99.4 shows a panel of diagnostic plots.

Figure 99.3 Residuals vs. Regressor

Figure 99.4 Fit Diagnostics

A trend in the residuals would indicate nonconstant variance in the data. The plot of residuals by predicted
values in the upper-left corner of the diagnostics panel in Figure 99.4 might indicate a slight trend in the
residuals; they appear to increase slightly as the predicted values increase. A fan-shaped trend might indicate
the need for a variance-stabilizing transformation. A curved trend (such as a semicircle) might indicate
the need for a quadratic term in the model. Since these residuals have no apparent trend, the analysis is
considered to be acceptable.

Polynomial Regression

Consider a response variableY that can be predicted by a polynomial function of a regressor variableX. You
can estimate� 0, the intercept;� 1, the slope due toX; and� 2, the slope due toX 2, in

Yi D � 0 C � 1X i C � 2X 2
i C � i

for the observationsi D 1; 2; : : : ; n.

Consider the following example on population growth trends. The population of the United States from 1790
to 2000 is �t to linear and quadratic functions of time. Note that the quadratic term,YearSq, is created in
the DATA step; this is done since polynomial effects such asYear*Year cannot be speci�ed in the MODEL
statement in PROC REG. The data are as follows:

data USPopulation;
input Population @@;
retain Year 1780;
Year = Year+10;
YearSq = Year * Year;
Population = Population/1000;
datalines;

3929 5308 7239 9638 12866 17069 23191 31443 39818 50155
62947 75994 91972 105710 122775 131669 151325 179323 203211
226542 248710 281422
;

proc reg data=USPopulation plots=ResidualByPredicted;
var YearSq;
model Population=Year / r clm cli;

run;

The DATA option ensures that the procedure uses the intended data set. Any variable that you might add to
the model but that is not included in the �rst MODEL statement must appear in the VAR statement.

The “Analysis of Variance” and “Parameter Estimates” tables are displayed in Figure 99.5.

Figure 99.5 ANOVA Table and Parameter Estimates

The ModelF statistic is signi�cant (F = 228.92,p < 0.0001), indicating that the model accounts for a
signi�cant portion of variation in the data. The R-square indicates that the model accounts for 92% of the
variation in population growth. The �tted equation for this model is

Population D � 2345:85C 1:29� Year

In the MODEL statement, three options are speci�ed: R requests a residual analysis to be performed, CLI
requests 95% con�dence limits for an individual value, and CLM requests these limits for the expected value
of the dependent variable. You can request speci�c100.1 � �/ % limits with the ALPHA= option in the
PROC REG or MODEL statement.

Figure 99.6 shows the “Output Statistics” table. The residual, its standard error, the studentized residual, and
Cook'sD are displayed for each observation. The studentized residual is the residual divided by its standard
error. Cook'sD is a measure of the change in the predicted values upon deletion of that observation from the
data set; hence, it measures the in�uence of the observation on the estimated regression coef�cients. The
studentized residuals and Cook'sD statistics are displayed in Figure 99.7.

Studentized residuals follow a beta distribution and can be used to identify outlying or extreme observations:

� Having a few large absolute studentized residuals is no cause for alarm.
� Many observations having absolute studentized residuals greater than two might indicate an inadequate

model.
� Absolute studentized residuals greater than three are rarer; they should be regarded as extreme, and the

data should be reviewed carefully.

Figure 99.6 Output Statistics

Figure 99.7 Studentized Residuals

Figure 99.8 shows the residual statistics table. A fairly close agreement between the PRESS statistic (see
Table 99.7) and the sum of squared residuals indicates that the MSE is a reasonable measure of the predictive
accuracy of the �tted model (Neter, Wasserman, and Kutner 1990).

Figure 99.8 Residual Statistics

Graphical representations are very helpful in interpreting the information in the “Output Statistics” table.
When ODS Graphics is enabled, the REG procedure produces a default set of diagnostic plots that are
appropriate for the requested analysis.

Figure 99.9 displays a panel of diagnostics plots. These diagnostics indicate an inadequate model:

� The plots of residual and studentized residual versus predicted value show a clear quadratic pattern.

� The plot of studentized residual versus leverage seems to indicate that there are two outlying data
points. However, the plot of Cook'sD distance versus observation number reveals that these two points
are just the data points for the endpoint years 1790 and 2000. These points show up as apparent outliers
because the departure of the linear model from the underlying quadratic behavior in the data shows up
most strongly at these endpoints.

� The normal quantile plot of the residuals and the residual histogram are not consistent with the
assumption of Gaussian errors. This occurs as the residuals themselves still contain the quadratic
behavior that is not captured by the linear model.

� The plot of the dependent variable versus the predicted value exhibits a quadratic form around the
45-degree line that represents a perfect �t.

� The “Residual-Fit” (or RF) plot consisting of side-by-side quantile plots of the centered �t and the
residuals shows that the spread in the residuals is no greater than the spread in the centered �t. For
inappropriate models, the spread of the residuals in such a plot is often greater than the spread of the
centered �t. In this case, the RF plot shows that the linear model does indeed capture the increasing
trend in the data, and hence accounts for much of the variation in the response.

